

```

/* Filename:      MLE9.LIM                                     */
/* Date:          07 July 1998                                 */
/* Project:       Determinants of Youth Employment           */
/* Written by:    Owen Gabbittas (Trade & Economic Studies Branch) */

/* Purpose:       Conducts SURE regressions using MLE         */
/*               with aggregated youth                         */
/*               and data in first differences                */
/*               and no intercepts                             */
/*               and without adult males                       */

Open; output=v:\youthemp\time\limdep\mle9.out $
Title; output file v:\..\mle9.out (first differences - no constant or
Am) $

Reset $

/* ==== Read in data - variable names in first line ==== */
Read; file = v:\youthemp\time\limdep\input4.wk1
      ; format = wks
      ; names = $

/* y - youth (aged 15 to 19) */
/* a - adults (aged 20 to 64) */
/* m - male */
/* f - female */
/* ie. afm - adult female */

/* Variables read from the input file in the following order: */
/* Industry Year dlnQ dlnr dlnWy dlnWam dlnWaf dEdy dEdam dEdaf */
/* dMy dMam dMaf dCy dCam dCaf dCk, dlnrWy, dlnrWam, dlnrWaf */

/* list; dCy, dWy, dEdy, dMy $ */
/* list; dCam, dWam, dEdam, dMam $ */
/* list; dCaf, dWaf, dEdaf, dMaf $ */
/* list; dCk, dlnr $ */
/* list; dCy, dlnWy, dlnWam, dlnWaf, dlnr ; file $ */
/* list; Year, Industry, dlnQ $ */

/* ===== Create industry dummy variables ===== */
/*
/*   A - Agriculture, forestry, fishing & hunting
/*   C - Manufacturing
/*   E - Construction
/*   F - Wholesale trade
/*   G - Retail trade (ommitted as biggest employer of youth)
/*   H - Accommodation, cafes & restuarants
/*   I - Transport, storage & communication services
/*   P - Cultural & personal services
/*   Indx represents the industry dummy for industry X
/*
/* ===== */

Create; if (Industry = 1) Inda = 1; (Else) Inda = 0
      ; if (Industry = 2) Indc = 1; (Else) Indc = 0
      ; if (Industry = 3) Inde = 1; (Else) Inde = 0
      ; if (Industry = 4) Indf = 1; (Else) Indf = 0
      ; if (Industry = 6) Indh = 1; (Else) Indh = 0
      ; if (Industry = 7) Indi = 1; (Else) Indi = 0

```

```

; if (Industry = 8) Indp = 1; (Else) Indp = 0 $

/* variables first differenced and logged in Excel owing to panel data
*/
/* r has already been converted to an index with base 1984=50
*/
Create ; dlnWk = dlnr $

/* list ; dlnWy, dlnWaf, dlnWk, dlnQ ; file $ */
Namelist ; Price = dlnWy, dlnWaf, dlnWk
; relPrice = dlnrWy, dlnrWaf
; Costshar = dCy, dCaf
; Educate = dEdy, dEdaf
; Ind = Inda, Indc, Inde, Indf, Indh, Indi, Indp $

/* ==== Seemingly unrelated regressions (SURE) - MLE ==== */

/* (a) Unconstrained */
Sure; LHS = Costshar
; Labels =
byy, byaf, byk, byq,
bafy, bafaf, bafk, bafq
; RHS = Price, dlnQ
; Pattern =
byy, byaf, byk, byq,
bafy, bafaf, bafk, bafq
; printvc $

Dstat ; Rhs = Price, dlnQ ; output = 2 $

/* (b) Imposing symmetry only */
Sure; LHS = Costshar
; Labels = byy, byaf, byk, byq, bafaf, bafk, bafq
; RHS = Price, dlnQ
; Pattern =
byy, byaf, byk, byq,
byaf, bafaf, bafk, bafq
; printvc $

Dstat ; Rhs = Price, dlnQ ; output = 2 $

/* (c) Imposing homogeneity only */
Sure; LHS = Costshar
; Labels = byy, byaf, byq, bafy, bafaf, bafq
; RHS = relPrice, dlnQ
; Pattern =
byy, byaf, byq,
bafy, bafaf, bafq
; printvc $

Dstat ; Rhs = relPrice, dlnQ ; output = 2 $

/* (d) Imposing symmetry & homogeneity only */
Sure; LHS = Costshar
; Labels = byy, byaf, byq, bafaf, bafq

```

```

; RHS = relPrice, dlnQ
; Pattern =
  byy, byaf, byq,
  byaf, bafaf, bafq
; printvc $

Dstat ; Rhs = relPrice, dlnQ ; output = 2 $

Stop $

/* ==== Incorporating additional environmental variables ==== */

/* (e) Industry dummy variables */
Sure; LHS = Costshar
; Labels = byy, byam, byaf, byq,
  bamam, bamaf, bamq,
  bafaf, bafq,
  Da, Dc, De, Df, Dh, Di, Dp
; RHS = relPrice, dlnQ, Ind
; Pattern =
  byy, byam, byaf, byq, Da, Dc, De, Df, Dh, Di, Dp,
  byam, bamam, bamaf, bamq, Da, Dc, De, Df, Dh, Di, Dp,
  byaf, bamaf, bafaf, bafq, Da, Dc, De, Df, Dh, Di, Dp $

/* (f) Industry dummy & environmental variables */
Sure; LHS = Costshar
; Labels = byy, byam, byaf, byq,
  bamam, bamaf, bamq,
  bafaf, bafq,
  Da, Dc, De, Df, Dh, Di, Dp,
  Ey, Eam, Eaf
; RHS = relPrice, dlnQ, Ind, Educate
; Pattern =
  byy, byam, byaf, byq, Da, Dc, De, Df, Dh, Di, Dp, Ey, Eam, Eaf,
  byam, bamam, bamaf, bamq, Da, Dc, De, Df, Dh, Di, Dp, Ey, Eam, Eaf,
  byaf, bamaf, bafaf, bafq, Da, Dc, De, Df, Dh, Di, Dp, Ey, Eam, Eaf
$

```