

```

/* Filename:      MLE7.LIM                                     */
/* Date:          07 July 1998                                 */
/* Project:       Determinants of Youth Employment            */
/* Written by:    Owen Gabbittas (Trade & Economic Studies Branch) */

/* Purpose:      Conducts SURE regressions using MLE         */
/*               with aggregated youth                         */
/*               and without adult males                      */

Open; output=v:\youthemp\time\limdep\mle8.out $
Title; output file v:\..\mle8.out (aggregated youth with no Am) $

Reset $

/* ==== Read in data - variable names in first line ==== */
Read; file = v:\youthemp\time\limdep\input2.wk1
      ; format = wks
      ; names = $

/* y - youth (aged 15 to 19) */
/* a - adults (aged 20 to 64) */
/* m - male */
/* f - female */
/* ie. afm - adult female */

/* Variables read from the input file in the following order:
*/
/* Industry Year Q r Wy Wam Waf Edy Edam Edaf My Mam Maf Cy Cam Caf Ck
*/

/* list; Cy, Wy, Edy, My $ */
/* list; Cam, Wam, Edam, Mam $ */
/* list; Caf, Waf, Edaf, Maf $ */
/* list; Ck, r */
/* list ; Wy, Wam, Waf, r ; file $ */
/* list; Year, Industry, Q $ */

/* ===== Create industry dummy variables ===== */
/*
/* A - Agriculture, forestry, fishing & hunting */
/* C - Manufacturing */
/* E - Construction */
/* F - Wholesale trade */
/* G - Retail trade (ommitted as biggest employer of youth) */
/* H - Acommodation, cafes & restuarants */
/* I - Transport, storage & communication services */
/* P - Cultural & personal services */
/* Indx respresents the industry dummy for industry X */
/*
/* ===== */

Create; if (Industry = 1) Inda = 1; (Else) Inda = 0
      ; if (Industry = 2) Indc = 1; (Else) Indc = 0
      ; if (Industry = 3) Inde = 1; (Else) Inde = 0
      ; if (Industry = 4) Indf = 1; (Else) Indf = 0
      ; if (Industry = 6) Indh = 1; (Else) Indh = 0
      ; if (Industry = 7) Indi = 1; (Else) Indi = 0
      ; if (Industry = 8) Indp = 1; (Else) Indp = 0 $

```

```

/* Deflate capital series 1989-90 = 50 instead of 100 */
Create ; defr = r / 2 $

/* Create relative wage terms - relative to the price of capital */
Create ; relWy = Wy / defr $
Create ; relWaf = Waf / defr $

/* list ; relWy, Wam, Waf ; file $ */

/* Create natural logs of the relative price terms */
Create ; LrelWy = log(relWy) $
Create ; LrelWaf = log(relWaf) $

/* list ; LrelWy, LrelWam, LrelWaf ; file $ */

/* Create natural logs of the absolute variables */
Create ; LWy = log(Wy) $
Create ; LWaf = log(Waf) $
Create ; LWk = log(defr) $
Create ; LQ = log(Q) $

/* list ; LWy, LWam, LWaf, LWk, LQ ; file $ */
Namelist ; Price = LWy, LWaf, LWk
          ; relPrice = LrelWy, LrelWaf
          ; Costshare = Cy, Caf
          ; Educate = Edy, Edaf
          ; Ind = Inda, Indc, Inde, Indf, Indh, Indi, Indp $

/* ==== Seemingly unrelated regressions (SURE) - MLE ==== */

/* (a) Unconstrained */
Sure; LHS = Costshare
      ; Labels =
        ay, byy, byaf, byk, byq,
        aaf, bafy, bafaf, bafk, bafq
      ; RHS = one, Price, LQ
      ; Pattern =
        ay, byy, byaf, byk, byq,
        aaf, bafy, bafaf, bafk, bafq $

Dstat ; Rhs = Price, LQ ; Output = 3 $

/* (b) Imposing symmetry only */
Sure; LHS = Costshare
      ; Labels = ay, aaf, byy, byaf, byk, byq, bafaf, bafk, bafq
      ; RHS = one, Price, LQ
      ; Pattern =
        ay, byy, byaf, byk, byq,
        aaf, byaf, bafaf, bafk, bafq $

/* (c) Imposing homogeneity only */
Sure; LHS = Costshare

```

```

; Labels = ay, byy, byaf, byq, aaf, bafy, bafaf, bafq
; RHS = one, relPrice, LQ
; Pattern =
ay, byy, byaf, byq,
aaf, bafy, bafaf, bafq $

Dstat ; Rhs = relPrice, LQ ; Output = 3 $

/* (d) Imposing symmetry & homogeneity only */
Sure; LHS = Costshar
; Labels = ay, byy, byaf, byq, aaf, bafaf, bafq
; RHS = one, relPrice, LQ
; Pattern =
ay, byy, byaf, byq,
aaf, byaf, bafaf, bafq $

/* ==== Incorporating additional environmental variables ==== */

/* (e) Industry dummy variables */
Sure; LHS = Costshar
; Labels = ay, byy, byaf, byq,
aaf, bafaf, bafq, Da, Dc, De, Df, Dh, Di, Dp
; RHS = one, relPrice, LQ, Ind
; Pattern =
ay, byy, byaf, byq, Da, Dc, De, Df, Dh, Di, Dp,
aaf, byaf, bafaf, bafq, Da, Dc, De, Df, Dh, Di, Dp $

/* (f) Industry dummy & environmental variables */
Sure; LHS = Costshar
; Labels = ay, byy, byaf, byq,
aaf, bafaf, bafq,
Da, Dc, De, Df, Dh, Di, Dp,
Ey, Eaf
; RHS = one, relPrice, LQ, Ind, Educate
; Pattern =
ay, byy, byaf, byq, Da, Dc, De, Df, Dh, Di, Dp, Ey, Eaf,
aaf, byaf, bafaf, bafq, Da, Dc, De, Df, Dh, Di, Dp, Ey, Eaf
$

```