
/* Filename:   OLS.LIM                                         */
/* Date:       26 June 1998                                    */
/* Project:    Determinants of Youth Employment                */
/* Written by: Owen Gabbitas (Trade & Economic Studies Branch) */

/* Purpose:    Conducts OLS regressions                        */

Open; output=v:\youthemp\time\limdep\ols.out $
Title; output file v:\..\ols.out $

Reset $

/* ==== Read in data - variable names in first line ==== */

Read; file = v:\youthemp\time\limdep\inputdat.wk1
    ; format = wks
    ; names = $

/* y - youth  (aged 15 to 19) */
/* a - adults (aged 20 to 64) */
/* m - male                   */
/* f - female                 */
/* ie. yf - female youth      */

/* list; Cym, Wym, Edym, Mym $ */
/* list; Cyf, Wyf, Edyf, Myf $ */
/* list; Cam, Wam, Edam, Mam $ */
/* list; Caf, Waf, Edaf, Maf $ */
/* list; Ck, r $ */
/* list; Year, Industry, Q $ */

Create; LWym=log(Wym)
      ; LWyf=log(Wyf)
      ; LWam=log(Wam)
      ; LWaf=log(Waf)

; LWk=log(r)
; lQ=log(Q)
; time = Trn(-13,0) $

/* ============ Create industry dummy variables ============= */
/*                                                            */
/*   A - Agriculture, forestry, fishing & hunting             */
/*   C - Manufacturing                                        */
/*   E - Construction                                         */
/*   F - Wholesale trade                                      */
/*   G - Retail trade (ommitted as biggest employer of youth) */
/*   H - Acommodation, cafes & restuarants                    */
/*   I - Transport, storage & communication services          */
/*   P - Cultural & personal services                         */
/*   Indx respresents the industry dummy for industry X       */
/*                                                            */
/* ========================================================== */

Create; if (Industry = 1) Inda = 1; (Else) Inda = 0
 ; if (Industry = 2) Indc = 1; (Else) Indc = 0
 ; if (Industry = 3) Inde = 1; (Else) Inde = 0
 ; if (Industry = 4) Indf = 1; (Else) Indf = 0
 ; if (Industry = 6) Indh = 1; (Else) Indh = 0
 ; if (Industry = 7) Indi = 1; (Else) Indi = 0
 ; if (Industry = 8) Indp = 1; (Else) Indp = 0 $



Namelist ; Wages = LWym, LWyf, LWam, LWaf
         ; Prices = LWym, LWyf, LWam, LWaf, LWk
         ; Costshar = Cym, Cyf, Cam, Caf
    ; Ed = Edym, Edyf, Edam, Edaf
    ; Ind = Inda, Indc, Inde, Indf, Indh, Indi, Indp $

/* ==== Simple OLS regressions of labour input demand equations ==== */

/* ==== Most basic model ===== */

/* Male youth */
Regress ; LHS = Cym
   ; RHS = one, LWym, LQ $

/* Female youth */
Regress ; LHS = Cyf
   ; RHS = one, LWyf, LQ $

/* Adult male */
Regress ; LHS = Cam
   ; RHS = one, LWam, LQ $

/* Adult female */
Regress ; LHS = Caf
   ; RHS = one, LWaf, LQ $

/* ==== Extended basic model (with additional price terms) ===== */

/* Male youth */
Regress ; LHS = Cym
   ; RHS = one, Prices, LQ $

/* Female youth */
Regress ; LHS = Cyf
   ; RHS = one, Prices, LQ $

/* Adult male */
Regress ; LHS = Cam
   ; RHS = one, Prices, LQ $

/* Adult female */
Regress ; LHS = Caf
   ; RHS = one, Prices, LQ $

/* ==== Expanded variable set ==== */

/* Male youth */
Regress ; LHS = Cym
   ; RHS = one, Prices, Ed, LQ, Ind, Time $

/* Female youth */
Regress ; LHS = Cyf
   ; RHS = one, Prices, Ed, LQ, Ind, Time $

/* Adult male */
Regress ; LHS = Cam



   ; RHS = one, Prices, Ed, LQ, Ind, Time $

/* Adult female */
Regress ; LHS = Caf
   ; RHS = one, Prices, Ed, LQ, Ind, Time $

/* ==== Base model with homogeneity ===== */

/* Male youth */
Regress ; LHS = Cym
   ; RHS = one, Prices, LQ
   ; Cls: B(2) + B(3) + B(4) +B(5) + B(6) = 1$

/* Female youth */
Regress ; LHS = Cyf
   ; RHS = one, Prices, LQ
   ; Cls: B(2) + B(3) + B(4) +B(5) + B(6) = 1$

/* Adult male */
Regress ; LHS = Cam
   ; RHS = one, Prices, LQ
   ; Cls: B(2) + B(3) + B(4) +B(5) + B(6) = 1$

/* Adult female */
Regress ; LHS = Caf
   ; RHS = one, Prices, LQ
   ; Cls: B(2) + B(3) + B(4) +B(5) + B(6) = 1$

Plot ; Lhs = Cym
     ; Rhs = Wym
     ; Title = Male_youth
     ; regression $


